NAG Library Routine Document

G13DBF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G13DBF calculates the multivariate partial autocorrelation function of a multivariate time series.

2 Specification

```
SUBROUTINE G13DBF (C0, C, LDC0, NS, NL, NK, P, V0, V, D, DB, W, WB, NVP,    &
                   WA, IWA, IFAIL)
INTEGER            LDC0, NS, NL, NK, NVP, IWA, IFAIL
REAL (KIND=nag_wp) C0(LDC0,NS), C(LDC0,LDC0,NL), P(NK), V0, V(NK),          &
                   D(LDC0,LDC0,NK), DB(LDC0,NS), W(LDC0,LDC0,NK),           &
                   WB(LDC0,LDC0,NK), WA(IWA)
```

3 Description

The input is a set of lagged autocovariance matrices $C_0, C_1, C_2, \ldots, C_m$. These will generally be sample values such as are obtained from a multivariate time series using G13DMF.

The main calculation is the recursive determination of the coefficients in the finite lag (forward) prediction equation

$$x_t = \Phi_{l,1} x_{t-1} + \cdots + \Phi_{l,l} x_{t-l} + e_{l,t}$$

and the associated backward prediction equation

$$x_{t-l-1} = \Psi_{l,1} x_{t-l} + \cdots + \Psi_{l,l} x_{t-1} + f_{l,t}$$

together with the covariance matrices D_l of $e_{l,t}$ and G_l of $f_{l,t}$.

The recursive cycle, by which the order of the prediction equation is extended from l to $l+1$, is to calculate

$$M_{l+1} = C_{l+1}^{\mathrm{T}} - \Phi_{l,1} C_l^{\mathrm{T}} - \cdots - \Phi_{l,l} C_1^{\mathrm{T}} \quad (1)$$

then $\Phi_{l+1,l+1} = M_{l+1} D_l^{-1}, \quad \Psi_{l+1,l+1} = M^{\mathrm{T}}{}_{l+1} G_l^{-1}$

from which

$$\Phi_{l+1,j} = \Phi_{l,j} - \Phi_{l+1,l+1} \Psi_{l,l+1-j}, \quad j = 1, 2, \ldots, l \quad (2)$$

and

$$\Psi_{l+1,j} = \Psi_{l,j} - \Psi_{l+1,l+1} \Phi_{l,l+1-j}, \quad j = 1, 2, \ldots, l. \quad (3)$$

Finally, $D_{l+1} = D_l - M_{l+1} \Phi^{\mathrm{T}}{}_{l+1,l+1}$ and $G_{l+1} = G_l - M^{\mathrm{T}}{}_{l+1} \Psi^{\mathrm{T}}{}_{l+1,l+1}$.

(Here T denotes the transpose of a matrix.)

The cycle is initialized by taking (for $l = 0$)

$$D_0 = G_0 = C_0.$$

In the step from $l = 0$ to 1, the above equations contain redundant terms and simplify. Thus (1) becomes $M_1 = C^{\mathrm{T}}{}_1$ and neither (2) or (3) are needed.

Quantities useful in assessing the effectiveness of the prediction equation are generalized variance ratios

$$v_l = \det D_l / \det C_0, \quad l = 1, 2, \ldots$$

and multiple squared partial autocorrelations

$$p_l^2 = 1 - v_l / v_{l-1}.$$

4 References

Akaike H (1971) Autoregressive model fitting for control *Ann. Inst. Statist. Math.* **23** 163–180

Whittle P (1963) On the fitting of multivariate autoregressions and the approximate canonical factorization of a spectral density matrix *Biometrika* **50** 129–134

5 Arguments

1: C0(LDC0, NS) – REAL (KIND=nag_wp) array *Input*

On entry: contains the zero lag cross-covariances between the NS series as returned by G13DMF. (C0 is assumed to be symmetric, upper triangle only is used.)

2: C(LDC0, LDC0, NL) – REAL (KIND=nag_wp) array *Input*

On entry: contains the cross-covariances at lags 1 to NL. $\mathrm{C}(i,j,k)$ must contain the cross-covariance, c_{ijk}, of series i and series j at lag k. Series j leads series i.

3: LDC0 – INTEGER *Input*

On entry: the first dimension of the arrays C0, C, D, DB, W and WB and the second dimension of the arrays C, D, W and WB as declared in the (sub)program from which G13DBF is called.

Constraint: $\mathrm{LDC0} \geq \max(\mathrm{NS}, 1)$.

4: NS – INTEGER *Input*

On entry: k, the number of time series whose cross-covariances are supplied in C and C0.

Constraint: $\mathrm{NS} \geq 1$.

5: NL – INTEGER *Input*

On entry: m, the maximum lag for which cross-covariances are supplied in C.

Constraint: $\mathrm{NL} \geq 1$.

6: NK – INTEGER *Input*

On entry: the number of lags to which partial auto-correlations are to be calculated.

Constraint: $1 \leq \mathrm{NK} \leq \mathrm{NL}$.

7: P(NK) – REAL (KIND=nag_wp) array *Output*

On exit: the multiple squared partial autocorrelations from lags 1 to NVP; that is, $\mathrm{P}(l)$ contains p_l^2, for $l = 1, 2, \ldots, \mathrm{NVP}$. For lags $\mathrm{NVP} + 1$ to NK the elements of P are set to zero.

8: V0 – REAL (KIND=nag_wp) *Output*

On exit: the lag zero prediction error variance (equal to the determinant of C0).

9: V(NK) – REAL (KIND=nag_wp) array *Output*

On exit: the prediction error variance ratios from lags 1 to NVP; that is, $\mathrm{V}(l)$ contains v_l, for $l = 1, 2, \ldots, \mathrm{NVP}$. For lags $\mathrm{NVP} + 1$ to NK the elements of V are set to zero.

10: D(LDC0, LDC0, NK) – REAL (KIND=nag_wp) array *Output*

On exit: the prediction error variance matrices at lags 1 to NVP.

Element (i, j, k) of D contains the prediction error covariance of series i and series j at lag k, for $k = 1, 2, \ldots, \text{NVP}$. Series j leads series i; that is, the (i, j)th element of D_k. For lags $\text{NVP} + 1$ to NK the elements of D are set to zero.

11: DB(LDC0, NS) – REAL (KIND=nag_wp) array *Output*

On exit: the backward prediction error variance matrix at lag NVP.

DB(i, j) contains the backward prediction error covariance of series i and series j; that is, the (i, j)th element of the G_k, where $k = \text{NVP}$.

12: W(LDC0, LDC0, NK) – REAL (KIND=nag_wp) array *Output*

On exit: the prediction coefficient matrices at lags 1 to NVP.

W(i, j, l) contains the jth prediction coefficient of series i at lag l; that is, the (i, j)th element of Φ_{kl}, where $k = \text{NVP}$, for $l = 1, 2, \ldots, \text{NVP}$. For lags $\text{NVP} + 1$ to NK the elements of W are set to zero.

13: WB(LDC0, LDC0, NK) – REAL (KIND=nag_wp) array *Output*

On exit: the backward prediction coefficient matrices at lags 1 to NVP.

WB(i, j, l) contains the jth backward prediction coefficient of series i at lag l; that is, the (i, j)th element of Ψ_{kl}, where $k = \text{NVP}$, for $l = 1, 2, \ldots, \text{NVP}$. For lags $\text{NVP} + 1$ to NK the elements of WB are set to zero.

14: NVP – INTEGER *Output*

On exit: the maximum lag, L, for which calculation of P, V, D, DB, W and WB was successful. If the routine completes successfully NVP will equal NK.

15: WA(IWA) – REAL (KIND=nag_wp) array *Workspace*

16: IWA – INTEGER *Input*

On entry: the dimension of the array WA as declared in the (sub)program from which G13DBF is called.

Constraint: $\text{IWA} \geq (2 \times \text{NS} + 1) \times \text{NS}$.

17: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: $\text{IFAIL} = 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, LDC0 < 1,
or NS < 1,
or NS $>$ LDC0,
or NL < 1,
or NK < 1,
or NK $>$ NL,
or IWA $< (2 \times \text{NS} + 1) \times \text{NS}$.

IFAIL $= 2$

C0 is not positive definite.

V0, V, P, D, DB, W, WB and NVP are set to zero.

IFAIL $= 3$

At lag $k = \text{NVP} + 1 \leq \text{NK}$, D_k was found not to be positive definite. Up to lag NVP, V0, V, P, D, W and WB contain the values calculated so far and from lag NVP $+ 1$ to lag NK the matrices contain zero. DB contains the backward prediction coefficients for lag NVP.

IFAIL $= -99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -399$

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -999$

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

The conditioning of the problem depends on the prediction error variance ratios. Very small values of these may indicate loss of accuracy in the computations.

8 Parallelism and Performance

G13DBF is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

G13DBF makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The time taken by G13DBF is roughly proportional to $\mathrm{NK}^2 \times \mathrm{NS}^3$.

If sample autocorrelation matrices are used as input, then the output will be relevant to the original series scaled by their standard deviations. If these autocorrelation matrices are produced by G13DMF, you must replace the diagonal elements of C_0 (otherwise used to hold the series variances) by 1.

10 Example

This example reads the autocovariance matrices for four series from lag 0 to 5. It calls G13DBF to calculate the multivariate partial autocorrelation function and other related matrices of statistics up to lag 3. It prints the results.

10.1 Program Text

```
    Program g13dbfe

!     G13DBF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
      Use nag_library, Only: g13dbf, nag_wp
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)               :: v0
      Integer                          :: i, ifail, iwa, k, ldc0, nk, nl, ns,  &
                                          nvp
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: c(:,:,:), c0(:,:), d(:,:,:),         &
                                          db(:,:), p(:), v(:), w(:,:,:),       &
                                          wa(:), wb(:,:,:)
!     .. Executable Statements ..
      Write (nout,*) 'G13DBF Example Program Results'
      Write (nout,*)

!     Skip heading in data file
      Read (nin,*)

!     Read series length, and numbers of lags
      Read (nin,*) ns, nl, nk

      ldc0 = ns
      iwa = (2*ns+1)*ns
      Allocate (c0(ldc0,ns),c(ldc0,ldc0,nl),p(nk),v(nk),d(ldc0,ldc0,nk),       &
        w(ldc0,ldc0,nk),wb(ldc0,ldc0,nk),wa(iwa),db(ldc0,ns))

!     Read autocovariances
      Read (nin,*)(c0(i,1:ns),i=1,ns)
      Read (nin,*)((c(i,1:ns,k),i=1,ns),k=1,nl)

!     Calculate multivariate partial autocorrelation function
      ifail = -1
      Call g13dbf(c0,c,ldc0,ns,nl,nk,p,v0,v,d,db,w,wb,nvp,wa,iwa,ifail)
      If (ifail/=0) Then
        If (ifail/=3) Then
          Go To 100
        End If
      End If

!     Display results
      Write (nout,99999) 'Number of valid parameters =', nvp
      Write (nout,*)
      Write (nout,*) 'Multivariate partial autocorrelations'
```

```
      Write (nout,99998) p(1:nk)
      Write (nout,*)
      Write (nout,*) 'Zero lag predictor error variance determinant'
      Write (nout,*) 'followed by error variance ratios'
      Write (nout,99998) v0, v(1:nk)
      Write (nout,*)
      Write (nout,*) 'Prediction error variances'
      Do k = 1, nk
        Write (nout,*)
        Write (nout,99997) 'Lag =', k
        Do i = 1, ns
          Write (nout,99998) d(i,1:ns,k)
        End Do
      End Do
      Write (nout,*)
      Write (nout,*) 'Last backward prediction error variances'
      Write (nout,*)
      Write (nout,99997) 'Lag =', nvp
      Do i = 1, ns
        Write (nout,99998) db(i,1:ns)
      End Do
      Write (nout,*)
      Write (nout,*) 'Prediction coefficients'
      Do k = 1, nk
        Write (nout,*)
        Write (nout,99997) 'Lag =', k
        Do i = 1, ns
          Write (nout,99998) w(i,1:ns,k)
        End Do
      End Do
      Write (nout,*)
      Write (nout,*) 'Backward prediction coefficients'
      Do k = 1, nk
        Write (nout,*)
        Write (nout,99997) 'Lag =', k
        Do i = 1, ns
          Write (nout,99998) wb(i,1:ns,k)
        End Do
      End Do

100   Continue

99999 Format (1X,A,I10)
99998 Format (1X,5F12.5)
99997 Format (1X,A,I5)
    End Program g13dbfe
```

10.2 Program Data

```
G13DBF Example Program Data
     4     5     3                                 :: NS,NL,NK
   .10900E-01 -.77917E-02  .13004E-02  .12654E-02
  -.77917E-02  .57040E-01  .24180E-02  .14409E-01
   .13004E-02  .24180E-02  .43960E-01 -.21421E-01
   .12654E-02  .14409E-01 -.21421E-01  .72289E-01 :: End of C0
   .45889E-02  .46510E-03 -.13275E-03  .77531E-02
  -.24419E-02 -.11667E-01 -.21956E-01 -.45803E-02
   .11080E-02 -.80479E-02  .13621E-01 -.85868E-02
  -.50614E-03  .14045E-01 -.10087E-02  .12269E-01
   .18652E-02 -.64389E-02  .88307E-02 -.24808E-02
  -.11865E-01  .72367E-02 -.19802E-01  .59069E-02
  -.80307E-02  .14306E-01  .14546E-01  .13510E-01
  -.21791E-02 -.29528E-01 -.15887E-01  .88308E-03
  -.80550E-04 -.37759E-02  .75463E-02 -.42276E-02
   .41447E-02 -.37987E-02  .19332E-02 -.17564E-01
  -.10582E-01  .67733E-02  .69832E-02  .61747E-02
   .41352E-02 -.16013E-01  .17043E-01 -.13412E-01
   .76079E-03 -.10134E-02  .11870E-01 -.41651E-02
   .36014E-02 -.36375E-02 -.25571E-01  .50218E-02
  -.13924E-01  .11718E-01 -.59088E-02  .59297E-02
```

```
  .10739E-01 -.14571E-01  .13816E-01 -.12588E-01
 -.64365E-03 -.44556E-02  .51334E-02  .71587E-03
  .63617E-02  .15217E-03  .27270E-02 -.22261E-02
 -.85855E-02  .14468E-02 -.28698E-02  .44384E-02
  .68339E-02 -.21790E-02  .13759E-01  .28217E-03  :: End of C
```

10.3 Program Results

```
G13DBF Example Program Results

Number of valid parameters =         3

Multivariate partial autocorrelations
     0.64498     0.92669     0.84300

Zero lag predictor error variance determinant
followed by error variance ratios
     0.00000     0.35502     0.02603     0.00409

Prediction error variances

Lag =    1
     0.00811    -0.00511     0.00159    -0.00029
    -0.00511     0.04089     0.00757     0.01843
     0.00159     0.00757     0.03834    -0.01894
    -0.00029     0.01843    -0.01894     0.06760

Lag =    2
     0.00354    -0.00087    -0.00075    -0.00105
    -0.00087     0.01946     0.00535     0.00566
    -0.00075     0.00535     0.01900    -0.01071
    -0.00105     0.00566    -0.01071     0.04058

Lag =    3
     0.00301    -0.00087    -0.00054     0.00065
    -0.00087     0.01824     0.00872     0.00247
    -0.00054     0.00872     0.00935    -0.00216
     0.00065     0.00247    -0.00216     0.02254

Last backward prediction error variances

Lag =    3
     0.00331    -0.00392    -0.00106     0.00592
    -0.00392     0.01890     0.00348    -0.00330
    -0.00106     0.00348     0.01003    -0.01054
     0.00592    -0.00330    -0.01054     0.03336

Prediction coefficients

Lag =    1
     0.81861     0.23399    -0.17097     0.09256
     0.06738    -0.48720    -0.14064     0.04295
     0.15036     0.11924    -0.36725    -0.42092
    -0.70971     0.02998     0.59779     0.34610

Lag =    2
    -0.34049    -0.13370     0.40610    -0.02183
    -1.27574    -0.13591    -0.65779    -0.11267
    -0.45439     0.19379     0.63420     0.33920
    -0.43237    -0.54848    -0.62897     0.16670

Lag =    3
     0.16437     0.13858     0.01290     0.03463
     0.39291     0.07407    -0.08802    -0.15361
    -1.29240    -0.24489     0.30235     0.39442
     0.89768    -0.39040     0.25151    -0.28304

Backward prediction coefficients

Lag =    1
```

```
      0.41541      0.06149      0.15319      0.05079
      0.12370     -0.26471     -0.22721      0.48503
     -0.86933     -0.47373      0.37924      0.13814
      1.30779     -0.09178     -1.45398     -0.21967

 Lag =     2
     -0.06740     -0.12255     -0.13673     -0.09730
     -1.24801      0.03090      0.51706     -0.28925
      0.98045     -0.20194      0.16307     -0.10869
     -1.68389     -0.74589      0.52900      0.41580

 Lag =     3
      0.03794      0.10491     -0.21635      0.08015
      0.75392      0.22603     -0.25661     -0.47450
     -0.00338      0.05636     -0.08818      0.12723
      0.55022     -0.41232      0.71649     -0.14565
```